
The included FAVES.WPM macro displays a menu of your frequently used files, keyboard merges or
macros on your screen. All you have to do is scroll to the item you want and press (Enter). If you select a
file, the macro retrieves the file to your screen. If you select a keyboard merge, the macro runs the
merge. If you choose a macro, the FAVES macro invokes the macro you've chosen.

Usage

To use the macro, press Macro (Alt-F10), type “faves” and press (Enter). The menu displays with your
customized list. Use the (Up Arrow) and (Down Arrow) keys to highlight the item you wish to use and
press (Enter). You can also press the corresponding mnemonic letter (i.e., the bolded letter) to activate
the item. Note that the items in the list are just placeholders. See “Customization” below to make the
macro your own.

Find Faves Faster!

For even quicker access to your most used files, merges and macros, rename the macro to an Alt-letter
macro. Do this by pressing List (F5), typing the full path to your macros directory — like c:\wp51\macros
— and pressing (Enter). Highlight FAVES.WPM and press (3) Move/Rename. Delete FAVES.WPM at the
New name: prompt and type “altx.wpm,” where “x” is your chosen letter name. Press (Enter) and Exit
(F7) to return to the document screen.

Customization

This macro requires extensive customization to work with your personal files. To do so, press (Ctrl-F10),
type “faves” and press 2 Edit. You are now in the macro editor and will see the macro commands as
listed below (and without the numbered lines on the left). The highlighted lines are those you need to
change.

For example, I have a favorite document called LTRMOM that I want add as the first item in the list. To
do so, I must change the macro in four places as :

  C First, I go to line 19 and change {^]}D{^\}ocument to {^]}L{^\}trmom.
  C I then go to line 29 and change d~Doc~D~Doc to l~Ltrmom~D~Ltrmom.
  C Next I go to line 48 and change 1~Doc to 1~Ltrmom.
  C Finally, I go to line 51 and change {LABEL}Doc~{CALL}CheckDoc~{Retrieve}Doc{Enter}{QUIT} to

{LABEL}Ltrmom~{CALL}CheckDoc~{Retrieve}Ltrmom{Enter}{QUIT}.

Now that you’re onto how to customize the macro, you can change it for your own favorite documents,
merges or macros. Just remember to make consistent changes to all four appropriate places for each
item.

The macro as seen in the macro editor (without the numbered lines):

1 {DESCRIPTION}Document Management Macro~
2 {;}************************************************************~
3 {;}As published in: WordPerfect Magazine~
4 {;}"Find Your Favorite Files Fast," May, 1994~
5 {;}(c) Copyright 1994 WordPerfect Corporation~
6 {;}************************************************************~
7
8 {ON CANCEL}{GO}Cancel~~
9 {ASSIGN}ItemNum~7~

10 {FOR}R~1~{VARIABLE}ItemNum~~1~
11 {ASSIGN}{VARIABLE}R~R~~
12 {END FOR}
13 {ASSIGN}1R~{^R}~{ASSIGN}Number~1~
14
15 {LABEL}Top~
16 {DISPLAY ON}
17 {CHAR}Option~
18 {^P}{NTOK}28~{NTOK}8~64444444444444444444447
19 {^P}{NTOK}28~{NTOK}9~5{VARIABLE}1R~       {^]}D{^\}ocument      {^S}5
20 {^P}{NTOK}28~{NTOK}10~5{VARIABLE}2R~       {^]}L{^\}etter        {^S}5
21 {^P}{NTOK}28~{NTOK}11~5{VARIABLE}3R~       {^]}M{^\}emo          {^S}5
22 {^P}{NTOK}28~{NTOK}12~5{VARIABLE}4R~       {^]}I{^\}nvoice       {^S}5
23 {^P}{NTOK}28~{NTOK}13~5{VARIABLE}5R~       {^]}F{^\}axBox        {^S}5
24 {^P}{NTOK}28~{NTOK}14~5{VARIABLE}6R~       C{^]}a{^\}lculator    {^S}5
25 {^P}{NTOK}28~{NTOK}15~5{VARIABLE}7R~       {^]}C{^\}ancel        {^S}5
26 {^P}{NTOK}28~{NTOK}16~94444444444444444444448
27 {^P}{NTOK}1~{NTOK}25~~
28 {CASE}{VARIABLE}Option~~
29 d~Doc~D~Doc~l~Letter~L~Letter~m~Memo~M~Memo~
30 i~Invoice~I~Invoice~f~FaxBox~F~FaxBox~a~Calculator~
31 A~Calculator~c~Cancel~C~Cancel~
32 {Down}~Down~{Up}~Up~{Enter}~Enter~{OTHERWISE}~Beep~~
33
34 {LABEL}Down~
35 {IF}{VARIABLE}Number~={VARIABLE}ItemNum~~{GO}Top~{END IF}
36 {ASSIGN}{VARIABLE}Number~R~~
37 {ASSIGN}Number~{VARIABLE}Number~+1~
38 {ASSIGN}{VARIABLE}Number~R~{^R}~{GO}Top~
39
40 {LABEL}Up~
41 {IF}{VARIABLE}Number~=1~{GO}Top~{END IF}
42 {ASSIGN}{VARIABLE}Number~R~~
43 {ASSIGN}Number~{VARIABLE}Number~-1~
44 {ASSIGN}{VARIABLE}Number~R~{^R}~{GO}Top~
45

46 {LABEL}Enter~
47 {CASE}{VARIABLE}Number~~
48 1~Doc~2~Letter~3~Memo~4~Invoice~5~FaxBox~6~Calculator~
49 7~Cancel~{OTHERWISE}~Top~~
50
51 {LABEL}Beep~{BELL}{GO}Top~
52 {LABEL}Doc~{CALL}CheckDoc~{Retrieve}Doc{Enter}{QUIT}
53 {LABEL}Letter~{CALL}CheckDoc~{Merge/Sort}1Letter{Enter}{Enter}{QUIT}
54 {LABEL}Memo~{CALL}CheckDoc~{Merge/Sort}1Memo{Enter}{Enter}{QUIT}
55 {LABEL}Invoice~{CALL}CheckDoc~{Merge/Sort}1Invoice{Enter}{Enter}{QUIT}
56 {LABEL}FaxBox~{Screen}{Screen}{NEST}FaxBox~{QUIT}
57 {LABEL}Calculator~{Screen}{Screen}{NEST}Calc~{QUIT}
58 {LABEL}Cancel~{Screen}{Screen}{QUIT}
59
60 {LABEL}CheckDoc~
61 {IF}{SYSTEM}Document~!=256~
62 {BELL}{PROMPT}{^R}Clear Screen, Rerun Macro{^S}~
63 {WAIT}30~{QUIT}
64 {END IF}
65 {RETURN}

